

Clase 0: Lectura Completa del Modelo GAMS (ps.gms)

Fase Cero - Tesis Portafolio DRL

13 de abril de 2026

Propósito de este Documento

Este apunte deconstruye línea por línea el archivo `ps.gms`, que constituye la implementación del modelo de optimización de portafolios con regímenes HMM en GAMS. Comprender cada instrucción es requisito obligatorio antes de (a) trasladar la lógica a DRL y (b) identificar qué parámetros son **inputs compartidos** y cuáles son **resultados exclusivos** del optimizador.

1. Bloque 0: Conjuntos y Alias

```
1 SETS
2   i   'activos'      / SPX, CMC200 /
3   k   'estados'      / bear, bull /
4   t   'periodos'     / t1*t163 / ;
5 ALIAS (i,j);
```

1.1. Explicación

- **Set i :** Define los dos activos del portafolio. `SPX` corresponde al índice S&P 500 y `CMC200` a un índice de las 200 criptomonedas de mayor capitalización.
- **Set k :** Los dos regímenes ocultos del modelo HMM. `bear` = mercado bajista (alta volatilidad) y `bull` = mercado alcista (baja volatilidad).
- **Set t :** Horizonte temporal de 163 periodos semanales.
- **ALIAS (i,j) :** Crea un clon del índice i para poder escribir sumas dobles como $\sum_{i,j}$ en las matrices de covarianza.

2. Bloque 1: Datos Exógenos (Tablas de Input)

2.1. 1.1 Probabilidades HMM

```
1 TABLE prob_spx(t,k)    'Probabilidades para SPX'
2                        bear      bull
3 t1                     0.098...  0.901...
4 t2                     0.914...  0.085...
5 ...
```

Estas tablas contienen la probabilidad de que cada activo se encuentre en régimen **bear** o **bull** en cada periodo t . Son el resultado de un modelo **Hidden Markov Model (HMM)** estimado previamente y de forma externa.

Notación formal: $P_i(k, t) = \Pr(\text{activo } i \text{ está en régimen } k \text{ en periodo } t)$.

Nota Crítica: Cada activo tiene sus propias probabilidades independientes. El SPX puede estar en **bull** mientras CMC200 está en **bear**.

2.2. 1.2 Retornos Semanales Observados

```
1 PARAMETER ret_semanal_spx(t) / t1 0.0214..., t2 -0.0330..., ... /;
2 PARAMETER ret_semanal_cmc200(t) / t1 0.0487..., t2 -0.0559..., ... /;
```

Estos son los retornos netos reales (históricos) de cada activo en cada periodo. Son **Ground Truth**: lo que efectivamente ocurrió en el mercado.

Notación formal: $R_{i,t}$ = retorno neto del activo i en periodo t .

2.3. 1.3 Consolidación de Parámetros

```
1 PARAMETER r(i,t), p(i,k,t);
2 r('SPX',t) = ret_semanal_spx(t);
3 r('CMC200',t) = ret_semanal_cmc200(t);
4 p('SPX',k,t) = prob_spx(t,k);
5 p('CMC200',k,t) = prob_cmc200(t,k);
```

Simplemente reorganiza las tablas anteriores en dos parámetros indexados genéricamente:

- $r(i, t)$: Retorno del activo i en t .
- $p(i, k, t)$: Probabilidad del activo i de estar en régimen k en t .

3. Bloque 2: Estimación de Momentos (El Corazón Estadístico)

Este es el bloque más importante de todo el archivo. Aquí se calculan las medias y covarianzas condicionales por régimen, y luego se mezclan para obtener los parámetros que alimentarán la función objetivo.

3.1. 2.1 Denominadores de Ponderación

```
1 den_mu(i,k) = sum(t, p(i,k,t));
2 den_sig(i,j,k) = sum(t, p(i,k,t)*p(j,k,t));
```

Qué hacen: Suman las probabilidades a lo largo de *todos* los periodos para crear un denominador de normalización.

Ecuaciones:

$$D_{i,k}^{\mu} = \sum_{t=1}^{163} P_i(k, t) \quad (1)$$

$$D_{i,j,k}^{\sigma} = \sum_{t=1}^{163} P_i(k, t) \cdot P_j(k, t) \quad (2)$$

ALERTA CRÍTICA (Hallazgo Post-Reunión): Estos denominadores usan $\sum_{t=1}^{163}$, es decir, **todos los periodos del dataset de golpe**. Esto significa que el $\hat{\mu}$ y $\hat{\sigma}$ que se calculan a continuación son estimaciones globales que miran hacia adelante (*look-ahead*). En un contexto DRL secuencial, esto constituye **data leakage** si se pasan tal cual. La solución del profesor es que para cada t , el $\hat{\mu}$ debe estimarse solo con datos $\{1, \dots, t\}$.

3.2. 2.2 Media Condicional por Régimen ($\hat{\mu}$)

```
1 mu_hat(i,k)$(den_mu(i,k) > 0) =
2   sum(t, p(i,k,t)*r(i,t)) / den_mu(i,k);
```

Ecuación:

$$\hat{\mu}_{i,k} = \frac{\sum_{t=1}^T P_i(k,t) \cdot R_{i,t}}{\sum_{t=1}^T P_i(k,t)} \quad (3)$$

Interpretación: Es un promedio ponderado de los retornos observados, donde el peso de cada observación es la probabilidad de estar en el régimen k . Por ejemplo, $\hat{\mu}_{SPX,bear}$ es el retorno promedio del SPX condicionado a las semanas en que el modelo HMM cree que el mercado estaba en crisis.

Resultado: Se obtiene un escalar $\hat{\mu}_{i,k}$ para cada combinación activo-régimen (4 valores en total: SPX-bear, SPX-bull, CMC200-bear, CMC200-bull).

PROBLEMA PARA DRL: Este $\hat{\mu}_{i,k}$ es calculado con las 163 semanas completas. En DRL, cuando la IA está en $t = 5$, solo debería conocer los datos de $t = 1$ a $t = 5$. Por tanto, necesitamos un $\hat{\mu}_{i,k}^{(t)}$ **rolling** (expansivo) que se recalcula en cada paso usando solo el pasado.

3.3. 2.3 Covarianza Condicional por Régimen ($\hat{\sigma}$)

```
1 sigma_hat(i,j,k)$(den_sig(i,j,k) > 0) =
2   sum(t, p(i,k,t)*p(j,k,t)
3     *(r(i,t)-mu_hat(i,k))*(r(j,t)-mu_hat(j,k))) / den_sig(i,j,k);
```

Ecuación:

$$\hat{\sigma}_{i,j,k} = \frac{\sum_{t=1}^T P_i(k,t) \cdot P_j(k,t) \cdot (R_{i,t} - \hat{\mu}_{i,k})(R_{j,t} - \hat{\mu}_{j,k})}{\sum_{t=1}^T P_i(k,t) \cdot P_j(k,t)} \quad (4)$$

Interpretación: Es la covarianza entre activos i y j condicionada al régimen k . Los retornos se centran usando el $\hat{\mu}_{i,k}$ calculado en el paso anterior, y se ponderan doblemente por las probabilidades de ambos activos.

Resultado: Una matriz 2×2 por cada régimen (bear y bull). En total, 8 valores (pero la simetría reduce a 6 únicos).

3.4. 2.4 Mezcla por Periodo: Los Famosos μ_{mix} y σ_{mix}

```
1 mu_mix(i,t) = sum(k, p(i,k,t)*mu_hat(i,k));
2 sigma_mix(i,j,t) = sum(k, p(i,k,t)*p(j,k,t)*sigma_hat(i,j,k));
```

Ecuaciones:

$$\mu_{mix}(i,t) = \sum_{k \in K} P_i(k,t) \cdot \hat{\mu}_{i,k} \quad (5)$$

$$\sigma_{mix}(i,j,t) = \sum_{k \in K} P_i(k,t) \cdot P_j(k,t) \cdot \hat{\sigma}_{i,j,k} \quad (6)$$

Interpretación: Aquí es donde se “mezclan” los dos mundos (bear y bull). Para cada periodo t , la media esperada del activo i es un promedio ponderado de su media-bear y su media-bull, donde los pesos son las probabilidades HMM de ese instante.

Ejemplo numérico: Si en t_5 , $P_{SPX}(bear) = 0,676$ y $\hat{\mu}_{SPX,bear} = -0,003$, $\hat{\mu}_{SPX,bull} = 0,012$:

$$\mu_{mix}(SPX, t_5) = 0,676 \cdot (-0,003) + 0,324 \cdot 0,012 = 0,00186$$

CONSECUENCIA: Dado que $\hat{\mu}_{i,k}$ es un escalar global (calculado con todos los T), el $\mu_{mix}(i, t)$ varía entre periodos **solo** porque $P_i(k, t)$ cambia, no porque $\hat{\mu}$ cambie. En la versión DRL correcta, tanto $P(k, t)$ como $\hat{\mu}^{(t)}$ deberían variar con t .

3.5. 2.5 Simetrización Numérica

```
1 sigma_tmp(i,j,t) = sigma_mix(i,j,t);
2 sigma_mix(i,j,t) = 0.5*(sigma_tmp(i,j,t) + sigma_tmp(j,i,t));
```

Operación estándar para garantizar que $\Sigma_{mix}(i, j, t) = \Sigma_{mix}(j, i, t)$. Corrige errores numéricos de punto flotante que podrían romper la simetría de la matriz de covarianza.

4. Bloque 3: Modelo de Optimización

4.1. 3.1 Escalares y Parámetros del Modelo

```
1 SCALARS
2   Capital_inicial / 10000 /
3   lambda_riesgo   / 0.10 /;
4 PARAMETER c_base(i);
5 c_base('SPX')     = 0.005;
6 c_base('CMC200') = 0.010;
7 PARAMETER w0(i);
8 w0('SPX')         = 0.5;
9 w0('CMC200')      = 0.5;
```

- **Capital_inicial** = \$10,000 USD.
- $\lambda = 0.10$ (Aversión al riesgo del inversor).
- $c_{SPX} = 0.5\%$ de costo por unidad de rotación en SPX.
- $c_{CMC200} = 1.0\%$ de costo por unidad de rotación (más caro por ser cripto).
- $w_0 =$ Portafolio inicial equitativo 50/50.

4.2. 3.2 Variables de Decisión

```
1 VARIABLES
2   z           "valor objetivo"
3   w(i,t)      "peso del activo i en t"
4   u(i,t)      "parte positiva del cambio de peso"
5   v(i,t)      "parte negativa del cambio de peso";
6
7 POSITIVE VARIABLES u, v;
8 w.lo(i,t) = 0; w.up(i,t) = 1;
```

- $w_{i,t} \in [0, 1]$: Fracción del capital en activo i en periodo t .
- $u_{i,t}, v_{i,t} \geq 0$: Variables auxiliares que linealizan el valor absoluto. Se cumple que $w_{i,t} - w_{i,t-1} = u_{i,t} - v_{i,t}$ y $|w_{i,t} - w_{i,t-1}| = u_{i,t} + v_{i,t}$.

4.3. 3.3 Restricciones

```

1 normalizacion_pesos(t)..  sum(i, w(i,t)) =e= 1;
2
3 rebalanceo_lineal(i,t)$(ord(t)>1)..
4   w(i,t) - w(i,t-1) =e= u(i,t) - v(i,t);
5
6 anclaje_inicial(i)..
7   w(i,'t1') - w0(i) =e= u(i,'t1') - v(i,'t1');

```

1. **Normalización:** $\sum_i w_{i,t} = 1 \quad \forall t$ (presupuesto completo).
2. **Rebalanceo:** Descompone el cambio de pesos en parte positiva y negativa para linealizar el valor absoluto.
3. **Anclaje:** El primer periodo arranca desde el portafolio inicial w_0 .

4.4. 3.4 Función Objetivo

```

1 FO_media_var_costo..
2   z =e= sum(t,
3       sum(i, w(i,t)*mu_mix(i,t))
4       - lambda_riesgo*sum((i,j), w(i,t)*w(j,t)*sigma_mix(i,j,t))
5       - sum(i, c_eff(i)*(u(i,t)+v(i,t))))
6   );

```

Ecuación formal:

$$\max_w Z = \sum_{t=1}^T \left[\underbrace{\sum_i w_{i,t} \mu_{mix}(i,t)}_{\text{Retorno esperado}} - \lambda \underbrace{\sum_{i,j} w_{i,t} w_{j,t} \sigma_{mix}(i,j,t)}_{\text{Riesgo}} - \underbrace{\sum_i c_i (u_{i,t} + v_{i,t})}_{\text{Costo transacción}} \right] \quad (7)$$

Observación clave: GAMS maximiza Z sobre **todos los periodos simultáneamente**. Es un problema de optimización estático que “ve el futuro”. El solver IPOPT encuentra los $w_{i,t}^*$ óptimos de golpe.

5. Bloque 4: Evaluación Ex-Post del Capital

```

1 LOOP(t$(ord(t)>1),
2   r_port_opt(t-1) = sum(i, w.l(i,t-1)*r(i,t-1));
3   turn_opt(i,t)   = u.l(i,t) + v.l(i,t);
4   cap_opt(t)      = cap_opt(t-1)*(1 + r_port_opt(t-1))
5                   - cap_opt(t-1)*sum(i, c_base(i)*turn_opt(i,t));
6 );

```

Este bloque simula cómo habría evolucionado el capital real si se hubieran ejecutado los pesos óptimos w^* :

$$x_{t+1} = x_t \cdot (1 + w_t^{*\top} R_t) - x_t \cdot \sum_i c_i \cdot |w_{i,t+1}^* - w_{i,t}^*| \quad (8)$$

Además calcula dos benchmarks naïve: un portafolio 50/50 con rebalanceo y otro buy&hold sin rebalanceo.

6. Bloque 5: Análisis de Sensibilidad

```

1 SET L / L1*L5 /, C / C1*C3 /;
2 lambda_grid('L1')=0.05; ... lambda_grid('L5')=1.00;
3 c_mult('C1')=0.5; c_mult('C2')=1.0; c_mult('C3')=2.0;
4
5 LOOP(L, LOOP(C,
6     lambda_riesgo = lambda_grid(L);
7     c_eff(i) = c_base(i)*c_mult(C);
8     SOLVE PortafolioEstadosCostos USING nlp MAXIMIZING z;
9     ... (recalcular capital) ...
10 ));

```

Ejecuta una grilla de $5 \times 3 = 15$ escenarios variando $\lambda \in \{0,05, 0,10, 0,20, 0,50, 1,00\}$ y un multiplicador de costos $\in \{0,5\times, 1,0\times, 2,0\times\}$, guardando el Z , capital final y retorno acumulado.

7. Resumen: Mapa de Parámetros para DRL

Parámetro	Compartir con DRL?	Nota
$P_i(k, t)$ (prob HMM)	Sí	Input exógeno idéntico
$R_{i,t}$ (retornos)	Sí	Ground Truth del mercado
$\hat{\mu}_{i,k}$ (media régimen)	Reformular	Debe ser rolling $\hat{\mu}^{(t)}$
$\hat{\sigma}_{i,j,k}$ (cov régimen)	Reformular	Debe ser rolling $\hat{\sigma}^{(t)}$
$\mu_{mix}(i, t)$	Reformular	Depende de $\hat{\mu}$ rolling
$\sigma_{mix}(i, j, t)$	Reformular	Depende de $\hat{\sigma}$ rolling
c_i, λ, w_0, x_0	Sí	Escalares compartidos
$w_{i,t}^*$ (pesos óptimos)	No	Resultado exclusivo de GAMS
Z^* (valor objetivo)	No	Resultado exclusivo de GAMS

Cuadro 1: Clasificación de parámetros para la transferencia GAMS $\rightarrow$ DRL.

Tarea Post-Reunión: El Data Handler de Python deberá implementar una versión **expanding-window** de las ecuaciones del Bloque 2, donde para cada periodo t , los momentos $\hat{\mu}_{i,k}^{(t)}$ y $\hat{\sigma}_{i,j,k}^{(t)}$ se calculan usando únicamente los datos $\{1, \dots, t\}$. Esto elimina el data leakage y genera un “ μ gorro” honesto para cada paso temporal del agente DRL.